Mathematics 244: Lab 1 Exact Solutions of Differential Equations

In this lab we use Maple to find exact solutions of differential equations and initial value problems and to help visualize solutions in cases when the solutions are only defined implicitly.

Please turn in only the printout of your Maple worksheet. Include explicit answers to all questions asked, using the **text** feature of Maple to insert them in the worksheet. Remove from the worksheet any extraneous material and any errors you have made.

Before you start up Maple, first copy the file lab1.mws into your directory from the Web page of the course. After you start up Maple, import this file into Maple by choosing **Open** from the **File** menu as described in the handout **Instructions for Use of Maple in Mathematics 244**. You will be asked to execute the Maple commands that have been placed in this worksheet and to add some of your own.

1a. Execute the following commands (already entered into Maple worksheet lab1.mws) which show how Maple solves the first order equation $dy/dx + y(x) = 1/(1 + e^x)$ (note that Maple uses `_C1` to denote an arbitrary constant), evaluates this solution at $x = 0$, and then determines the constant `_C1` so that $y(0) = c$. In this problem, the solution y is given explicitly as a function of x. Note that the commands `with(plots): with(DEtools):` must be entered to allow you to use the Maple commands which are part of these special Maple packages.

```
with(plots): with(DEtools):
de1:= diff(y(x),x) + y(x) = 1/(1 + exp(x));
s1:=dsolve(de1,y(x));
y0:=simplify(subs(x=0,rhs(s1)));
c1:=solve(y0=c,_C1);
s2:=subs(_C1=c1,s1);
```

This is the same procedure you would use to solve the inital value problem if you were not using Maple. Execute the following command, which shows how Maple can solve the initial value problem more simply and directly (note the use of braces {}).

```
s3:=dsolve({de1,y(0)=c},y(x));
```

The solution produced by Maple is really a family of solutions, i.e., there is a different solution corresponding to each particular choice of the constant c. Execute the following sequence of commands which defines particular solutions corresponding to the choices $c = -2, -1, 0, 1, 2$, and plots all the solutions on the same set of axes.

```
t1:=rhs(dsolve({de1,y(0)=-2},y(x)));
t2:=rhs(dsolve({de1,y(0)=-1},y(x)));
t3:=rhs(dsolve({de1,y(0)=0},y(x)));
t4:=rhs(dsolve({de1,y(0)=1},y(x)));
t5:=rhs(dsolve({de1,y(0)=2},y(x)));
plot({t1,t2,t3,t4,t5},x=-1..5);
```

To evaluate the solution `t1` at the point $x = 1.2$, type `subs(x=1.2,t1);` or to get a numerical value for this quantity type `evalf(subs(x=1.2,t1));`

1b. Note that in this case, no matter what initial condition we start with, all solutions tend to the same value as $x \to \infty$. What value is this?

1c. The qualitative behavior of the differential equation can also be determined by looking at a plot of its direction field. Execute the following statements which show the use of Maple's `dfieldplot` command to plot direction fields. This can also be combined with the `display` command to plot the direction field and solutions of the differential equation on the same set of axes. Note that the `dfieldplot` command assumes the differential equation is written in the form $dy/dx = f(x, y)$. Also note that when naming the output of a plot, changing the ending semicolon to a colon will suppress unwanted output.

```
de1m:= diff(y(x),x) = - y(x) + 1/(1 + exp(x));
dfieldplot(de1m,y(x), x=-1..5,y= -6..4,color=black);
df:= dfieldplot(de1m,y(x), x=-1..5,y= -6..4,color=black):
ds:= plot({t1,t2,t3,t4,t5},x=-1..5):
display({df,ds});
```

2a. Use Maple to find the general solution of the differential equation

$$x\frac{dy}{dx} + xy = 1 - y$$

and then the solution of the initial value problem consisting of this equation and the initial condition $y(1) = 1/2$. Plot the solution of the initial value problem over the interval $0 \le x \le 5$.

2b. Plot the direction field for the equation in part (a) over the range $0.1 \le x \le 5$, $-2 \le y \le 1$ and then plot the direction field and the solution found in part (a) on the same set of axes. Note: the differential equation must be rewritten in the form $dy/dx = f(x, y)$ for the correct use of `dfieldplot`.

3a. Execute the following commands in which Maple solves the differential equation

$$\frac{dy}{dx} = \frac{x - e^{-x}}{y + e^{y}}.$$

Note in this case, the solution $y(x)$ is only defined implicitly as a function of x, i.e., it appears in an equation involving both y and x in which y is not solved explicitly as a function of x.

```
de:=diff(y(x),x) = (x-exp(-x))/(y(x) + exp(y(x)));
soln:=dsolve(de,y(x));
```

If the solution is given in the special form $f(x, y) = constant$, one way to graph this family of solutions is to use the `contourplot` command. We introduce the name `ValC` for the expression which is constant using

```
ValC:=lhs(soln);
```

If the expression `soln` is not in this special form, you should examine it to find the constant introduced into the solution — it is likely to be called `_C1` — and solve for it. The expression that accomplishes this is

```
ValC:=solve(soln,_C1);
```

In any case, the graph is obtained from

```
contourplot(ValC,x=-2..4,y=0..4);
```

3b. Consider the initial condition $y(0) = 1$. Execute the following sequence of commands by which Maple determines the value of the constant `_C1` corresponding to the initial condition $y(0) = 1$, naming the result `c`, and then replaces the arbitrary constant `_C1` in the solution by this value. Finally, the `implicitplot` command is used to draw the graph of the solution of the initial value problem.

```
c:=simplify(subs({x=0,y(x)=1},ValC));
solns:=subs(_C1=c,soln);
implicitplot(solns,x=-2..4,y=0..4);
```

3c. The implicit solution obtained in part (b) should have a form equivalent to

$$\frac{1}{2}(y(x))^2 + e^{y(x)} - \frac{1}{2}x^2 - e^{-x} = -\frac{1}{2} + e.$$

To find the value of the solution $y(x)$ at say $x = 2$, we substitute $x = 2$ in the above equation and then determine the number y which satisfies the resulting equation

$$y^2/2 + e^y - 2 - e^{-2} = -1/2 + e.$$

Use Maple's `fsolve` command to solve the above equation to obtain the value $y(2)$.

3d. Although the solution of the initial value problem is only defined implicitly, Maple's `dsolve` command may sometimes return what appears to be an explicit solution. Execute the sequence of commands below to see how Maple uses the `RootOf` function to write the solution.

```
so:= dsolve({de,y(0)=1},y(x));
evalf(subs(x=0,rhs(so)));
evalf(subs(x=2,rhs(so)));
plot(rhs(so),x=-2..4);
```

The result may be surprising. The `RootOf` function is not guaranteed to select the root that you are thinking of.

4. Use Maple's `dsolve` command to find the general solution of the differential equation

$$\frac{dy}{dx} = \frac{1}{1+e^y}.$$

If you don't understand the expression that Maple gives you for the answer, use the *Help* menu. Then, use appropriate commands to plot a family of solutions over the range $-5 \le x \le 5$, $-2 \le y \le 2$.